NASA/ASF Project Report

Ref: NRA 94-MTPE-05
NASA Ref: 1995-ADRO00194
RTOP: 665-21-02
ADRO Project ID: 164

"Calibration Comparison of RSI and ASF Antarctic ScanSAR Narrow B Products"

Mark R. Drinkwater and Xiang Liu
Jet Propulsion Laboratory
California Institute of Technology

November 1999

Abstract

A meeting was held at JPL in summer 1999 between ASF staff and Mark Drinkwater to try to resolve the remaining problems in effective use and implementation of ASF Narrow ScanSAR (SNB) products. In converging on a solution it was resolved to (i) evaluate RSI-delivered SNB geocoded products for their calibration accuracy (ii) evaluate Jeremy Nicoll's post-processing calibration routine for ASF SNB products and (iii) compare each product with known sigma-naught values extracted from ERS-2 Scatterometer images of a stable fast ice/ice-shelf distributed target in Antarctica. This brief White Paper details the results and findings of this work.

These caveats apply to this report:

(1) A region of fast ice/ice shelf was chosen on the eastern side of the Antarctic peninsula on RADARSAT orbit 5994. It was selected for its 'apparent' uniformity of sigma-naught at 40^{o} incidence angle in the RSI and ASF products, and because it displays characteristics typical of a distributed target.

(2) There is a lack of *a priori* information about hh- versus vv-polarized returns from Antarctic fast ice. A distributed "volume-scattering" target was chosen because previous radar polarimetric experience shows that the response from volume scattering targets is typically depolarized to the extent that vv- and hh-pol returns are of similar magnitude. There is no detailed information however about the surface roughness of the chosen target, nor indeed whether there is significant anisotropy in that roughness.

(3) The ERS-2 image sigma-naught values are a temporal mean, extracted from an image comprised of measurement cells on multiple orbit passes. The quoted standard deviations are derived from the scatter of the individual instantaneous measurements obtained on each of the orbit passes over the averaging interval. Given the +/- 0.3 dB calibration accuracy of the ERS-2 scatterometer, these provide some understanding of the coherence time scales of the chosen target. Furthermore, the mean values in the target area are the only independent "control" for the evaluation of the relative accuracy of the ASF and RSI products.

(4) Problems with the ASF SNB leader files prevented precise positioning of the target region selected in the geocoded RSI products. Hence it was deemed necessary to shift the target region slightly in space to gain an appreciation of the uniformity of the chosen target.

Preparing the Data

RSI CDPF SNB Products

The RSI delivered ScanSAR Narrow B data products were processed by the Canadian Data Processing Facility (CDPF). These CEOS products were not simple or intuitive to understand and contain significant differences from CEOS product records generated by either ASF and the DLR (from personal experience). Rick Guritz provided a simple dd tape-reading script for decoding the individual data files. However, this did not work correctly as the RSI trailer file contains mixed records of variable block files. The trailer file is the most critical file to correctly decode, as it contains all the calibration information for the CSA SNB products. The leader file in contrast was the least problematic to decode but it does not contain the "meat" of the facility records required to calibrate the data.

Rick eventually was able to use "dump-multi_volume" to correctly decode the trailer file, and sent a revised trailer file for the scene of interest over the Larsen Ice Shelf. This approach did not successfully decode other files from the tape, and so it was necessary to reconstitute the files read using dd, with the trailer file successfully read using "dump-multi_volume".

It was then possible to extract the necessary fields from the trailer file and the image data file to do the conversion to Sigma Naught (by the rules laid out in the "unofficial" product specifications that found their way onto our desks). The effort of trying to understand the relevant details in the trailer and data file specifications was nevertheless time consuming, as the image has different metadata header length from that at ASF, and the trailer file fields (facility data and calibration records) were unintelligible for the most part. We took the time to write routines to extract the individual byte sequenced fields to obtain the necessary calibration constants to be able to calibrate the images. These PVWave routines are available in the Appendix.

ASF SNB Products

Originally "Unqualified" SNB products were processed and delivered to us in geocoded form. However it was determined that the framing was incorrect, and the original data was rescanned and framed using the new software. Unfortunately the scenes chosen from the catalog which were reprocessed for this test were not geocoded. Jeremy Nicoll had calibrated the ground-range images and supplied them by ftp for frames. The data files supplied were as follows;

R105994730G6S011
R105994730G6S012

The first problem noticed is that the ASF leader files claim that they are geocoded as UTM projected images, whereas the file names themselves appear to indicate 400m pixel spacing, ground-range images. This appears incorrect and is the first obvious problem. Furthermore, these data could not be geocoded using ASF software as a consequence. Thus, crude geocoding was implemented by rotating a subimage to match the target region in the RSI image. Mountain features evident in the scenes suggest that there is no critical difference in terms of distortion of the target of interest.

Target extraction and calibration of image DN values

The target site was chosen in the know geolocated RSI image at a uniform location corresponding to an incidence angle of 40^{o}. This incidence angle was chosen to allow extraction of normalized incidence angle backscatter data from simultaneous, overlapping ERS-2 wind scatterometer data. Furthermore, the target region lay near to the northern part of the Larsen Ice Shelf, in a uniform featureless region of fast ice. It turned out that only RSI frame 'C0014517.dat' overlaps the ice shelf region of interest.

The absolute range of incidence angles in RSI image frame 'C0014517.dat', (in the sample range of 1 to 11776 pixels per line) is 32.6 to 49 degrees. The pixel sample number corresponding to the incidence angle of 40 degrees is 5407 for (determined using the user-developed PVWave/IDL routine incidence.pro).



Figure 1. Target region in RSI data, selected from scene C0014517 on a fast ice region adjacent to the Larsen Ice Shelf. Region selected is 100 x 100 pixels in extent, and is centered at 40 degrees incidence angle for direct comparison with ERS scatterometer data.

The box identified in Figure 1 (the RSI data's pixel spacing is 25m) is sub-setted from a strip centered at 40^{o} incidence angle. This strip was first extracted from the full-swath data and rotated 90^{o} counter-clockwise for ease of viewing. The subarray of pixels was converted to calibrated sigma-naught values by use of a user-developed routine "tosigma0.pro" which converts the digital values of the chosen target region to calibrated backscatter coefficients (see Appendix for conversion routine). The required input parameter is the starting range pixel (sample number) for the 1st column of the box and the output data is a binary floating point array.

Results

A previous SNB geocoded image of the target region obtained from ASF allowed us to obtain the location of the extracted box in Figure 1. The lower left corner of the region of interest is -64.58^{o} S -60.30^{o} W.

ERS-1/2 Scatterometer Data

The latitude and longitude geographic coordinates derived from the ASF geolocated product corresponds to (i,j) pixel coordinates (209,643) in the ERS-1/2 SIRF grid. The DN number at this pixel (209,643) is 149 and 150 for both SIRF image 'ers2-a-Ant96-363-003.byt' and 'ers2-a-Ant96-360-000.byt' respectively. These two images are 6 day averages constructed at 3 day intervals and the converted mean dB values for the target location in each scene are -10.6 dB and -10.5 dB, respectively. The corresponding pixel standard deviations are 0.74 dB and 0.62 db, and are retrieved from statistics compiled from all overlapping cells on orbits crossing the target region during the 6 day imaging interval. The calibration accuracy of the ERS-2 data is 0.2 dB [Hawkins et al., 1999], which implies an extremely small temporal variability in the sigma-naught values. Furthermore, the mean values indicate a change of 0.1 dB over a several day period. This suggests that the calibrated backscatter of the selected distributed target varies by a small value which is negligible with respect to the calibration accuracy of either the RSI or ASF Radarsat products. An additional consideration is that a large variance in the inter-swath measurements would imply a azimuthal dependency in the sigma-naught values. The fact that the standard deviation is small implies that the target conforms to the hypothesized assumption of a distributed target.

ASF Post-Calibrated RADARSAT Product

The ASF product target region (6 x 6 pixels at 400 m resolution) yields a mean backscatter value of Sigma0 = -11.7278 dB with a standard deviation of 0.537513 dB. If we shift the box location by 2 pixels in each direction (to investigate variability resulting from possible geolocation inaccuracy) then we obtain small variations in the mean and standard deviation:

Target window position	Sigma0	Std
centered	-11.73	0.54
shifted to right	-12.22	1.02
shifted to left	-11.78	0.51
shifted to top	-12.07	0.80
shifted to bottom	-11.71	0.68

RSI Calibrated Product

If DN_j is the digital number which represents the magnitude of the j th. pixel from the start of a range line in the detected image data, then the corresponding value of radar brightness, $ß^o_j$, for the pixel is given by:

$$ß^o_j = 10*\log_{10}[(DN_j^2 + A3)/A2_j] \ \ dB \qquad 1$$

The value of $ß^o_j$ is first extracted using the routine Beta0.pro described in the Appendix. $A2_j$ is the scaling gain value for the j th. pixel, and A3 is the fixed offset (*which in the case of our image is zero). A3 is obtained directly from field 531 (offset) in the Radiometric Data Record. $A2_j$ is obtained by linear interpolation of the gain values (from lookup_tab) given in fields 16-527 of the Radiometric Data Record, as obtained using the routine in the Appendix. Interpolation of LUT Values is necessary to find $A2_j$.

The relationship between radar brightness ($ß^o$) and radar backscatter coefficient (σ^o) is:

$$\sigma^o_j = ß^o_j + 10*\log_{10}(\sin I_j) \ \ dB \qquad 2$$

where I_j is the incidence angle at the j th. range pixel. This formula assumes that the earth is a smooth ellipsoid at sea level.

Data required for calculating the incidence angle for any given pixel in the image range line are available in the CEOS record. Given these data, a relatively straightforward approximation to the incidence angle can be performed using the function incidence.pro provided in the Appendix. For scene products the resulting error in conversion from beta nought to sigma nought resulting from the approximation should be less than 0.4 dB (according to the RSI data product specifications).

For RSI product, the pixels within the target region yield a calibrated mean hh-polarized backscatter value of sigma0 = -17.41 dB and a standard deviation of 2.67 dB. The larger standard deviation is accounted for by the lower number of "effective looks" in these data and the fact that 100 x 100 pixels were averaged at 25 m resolution. ASF data are 400 m resolution data and there is a expected 16 fold [i.e. sqrt(16 * 16)] increase in standard deviation in the CDPF as a consequence of the pixel scale alone.

Conclusions

Unfortunately, a true direct comparison was not possible, since quite different products were provided by ASF for the purpose of this comparison study. The RSI-delivered CDPF data were 25m resolution geocoded data, and the ASF products were 400m non-geocoded products. This resulted in a difference in target size by a factor of 16 in each dimension, and thus a quite different number of effective looks in the data Notwithstanding these differences, the comparison between calibrated vv-polarized target data derived from ERS-2 scatterometer and hh-polarized ASF RADARSAT SAR pixel values shows a close correspondence. A mean sigma-naught of -10.6 ± 0.74 dB was obtained from the ERS-2 Wind Scatterometer and a calibrated value of -11.73 ± 0.54 dB from the ASF (Nicolls) post processed Narrow ScanSAR RADARSAT product. This difference of ~0.9 dB may be accounted for by slight geolocation errors in the positioning of the Antarctic target region, together with inaccuracies in the post-processing calibration routine. Furthermore, there may be physical reasons why the sample region does not conform as a perfect distributed target, thus accounting for slightly larger vv-polarized backscatter values.

Our result from the RSI data calibration appears to signify a considerably lower mean value of -17.41 ± 2.67 dB. The difference of -5 dB appears in the mean calibrated backscatter coefficient after converting pixel DN values from beta-naught to sigma-naught. The increased variance in the individual pixels may be accounted for by the different number of effective looks in the RSI data, coupled with the difference in pixel scale (25 m in the RSI product). Further independent checks may have to be performed with these data in order to shed some light on this difference.

Previous independent comparisons between RSI processed and ASF ScanSAR Wide B data have indicated that ASF calibration resulted in values which were consistently 3 dB lower than RSI calibrated values at high incidence angles [Vachon et al., 1999]. Our result indicates the opposite tendency, with ASF data providing sigma-naught values both higher than the RSI calibrated values and more consistent with the control region in the ERS-2 scatterometer data.

The general conclusion is that the ASF post-processed calibrated Narrow ScanSAR appear to give realistic pixel values, of sufficient accuracy for their extended use in further geophysical studies of Antarctic ice. However, a suitable post-processing SNB calibration routine must be developed which retains the maximum flexibility in terms of subsequent application of other ASF software routines. Furthermore, there are residual problems with the information contained in the leader file of ScanSAR Narrow B products which must be corrected if the archived data are to be correctly or successfully in scientific studies.

If RSI data is to be used in the future as a substitute for ASF product calibration and qualification, it is particularly important to read the trailer file correctly since it contain all the essential information for

calibration of the data. The RSI draft product specifications documentation is so general (since it was developed to apply to all product formats) that it is almost impossible with certainty whether we are performing the calibration in the correct manner.

References

Hawkins, R.K., E. Attema, R. Crapolicchio, P. Lecomte, J. Closa, P.J. Meadows, and S.K. Sristava, Stability of Amazon Backscatter at C-band: Spaceborne Results from ERS-1/2 and RADARSAT-1, *Proc. CEOS Working Group on Calibration and Validation SAR Workshop*, 26-29 October, 1999 Toulouse, France, 1999.

Vachon, P.W., J. Wolfe, and R.K. Hawkins, The Impact of RADARSAT ScanSAR Image Quality on Ocean Wind Retrieval, *Proc. CEOS Working Group on Calibration and Validation SAR Workshop*, 26-29 October, 1999 Toulouse, France, 1999.

APPENDIX

The following PVWave programs were used to extract calibrated pixel values from the RSI products;

The routine rd_rsi_tlr.pro is used to read the trailer file calibration parameter values required for the subsequent conversion of DN values to sigma naught.

```
PRO rd_rsi_tlr
;; ================================================================
;; This procedure extract fields from the Trailer file of  RSI's
;; RadarSat ScanSar data.
;;
;; Note: the trailer file 'rsi_dat.tlr' was provided by Rick  Guritz
;; who used 'dump_multi_volume' to stage the tape onto disk.
;;
;; Oct. 19, 1999  Xiang Liu
;; ================================================================

common data_sum,ellip_maj,ellip_min,pro_lat,plat_lat,pix_spacing
common proc_param,eph_orb_dat,srgr_coef
common radiometric,n_samp,samp_inc,lookup_tab,A3
common img_data,acc_year,acc_day,acc_msec,lat_mid

bufl=500000

openr,1,'/rime4/rad1/rsi/rsi_dat.tlr'
a=assoc(1,bytarr(bufl),0)
dat=a(0)

; from Dataset Summary Record
offset=720

ellip_maj=float(string(dat(180+offset:195+offset)))
ellip_min=float(string(dat(196+offset:211+offset)))
pro_lat=float(string(dat(116+offset:131+offset)))
plat_lat=float(string(dat(452+offset:459+offset)))
pix_spacing=float(string(dat(1702+offset:1717+offset)))

;info,ellip_maj,ellip_min,pro_lat,plat_lat,pix_spacing

; from Detailed Processing Parameters Record
offset=720l+4096+8960l+26500l

eph_orb_dat=float(string(dat(4648+offset:4663+offset)))

;for j=0,19 do begin  ;for choosing the closest time to acquisition time
for j=0,0 do begin
    m1=j*117l
    srgr_update=string(dat(4886+offset+m1:4906+offset+m1))
;    print,'SRGR update date/time:',srgr_update
```

```
    mm=4906+offset+m1

    srgr_coef=fltarr(6)
    for i=0,5 do begin
        k1=mm+1+i*16
        k2=k1+15
        srgr_coef(i)=float(string(dat(k1:k2)))
    endfor
endfor

;info,eph_orb_dat,srgr_update

; from Radiometric Data Record
offset=720l+4096+8960l+43186l

n_samp=string(dat(60+offset:67+offset))
samp_inc=string(dat(84+offset:87+offset))

lookup_tab=fltarr(512)
for i=0,512-1 do begin
    k1=88+offset+i*16
    k2=k1+15
    lookup_tab(i)=float(string(dat(k1:k2)))
endfor

A3=float(string(dat(8316+offset:8331+offset)))

;info,n_samp,samp_inc,A3

close,1

; from the Processed Data Record
openr,1,'/rime4/rad1/rsi/Data/C0014517.dat'

offset=16252     ;skip image file descriptor record
skiparr1=bytarr(offset)
readu,1,skiparr1

;while (NOT EOF(1)) do begin  ;to find out no.  of lines in the image

rec_seq=lonarr(1)
readu,1,rec_seq
;print,'Record sequence number:',rec_seq

arr4=bytarr(4)
readu,1,arr4

length=lonarr(1) & line_num=length & rec_num=length
n_left_pixel=length & n_data_pixel=length & n_right_pixel=length
readu,1,length
readu,1,line_num
readu,1,rec_num
readu,1,n_left_pixel
readu,1,n_data_pixel
readu,1,n_right_pixel
;print,'Length of pixel count each line:',length
```

```
;print,'Length of data pixel count each line:',n_data_pixel

readu,1,arr4

acc_year=lonarr(1) & acc_day=acc_year & acc_msec=acc_year
readu,1,acc_year        ;acquisition year
readu,1,acc_day         ;acquisition day of year
readu,1,acc_msec        ;acquisition msecs of day

skiparr2=bytarr(88)
readu,1,skiparr2

lat_mid=lonarr(1)
readu,1,lat_mid         ;millionths of degree
lat_mid=lat_mid/1.E+06  ;degree

;print,'Acquisition date: '+string(acc_year,format='(I4)')+'-'+  $
;      string(acc_day,format='(I3)')+'-'+string(acc_msec/1000.)+'sec'

skiparr3=bytarr(52+n_data_pixel(0))  ;read through the rest of line
readu,1,skiparr3

;end   ;end while

close,1

END
```

The printed output from the above routine is as follows;

```
ELLIP_MAJ (DATA_SUM) FLOAT     =       6378.14
ELLIP_MIN (DATA_SUM) FLOAT     =       6356.75
PRO_LAT (DATA_SUM) FLOAT     =      -66.0631
PLAT_LAT (DATA_SUM) FLOAT     =      -67.4010
PIX_SPACING (DATA_SUM) FLOAT     =       25.0000
EPH_ORB_DAT (PROC_PARAM) FLOAT     =   7.16705e+06
SRGR_UPDATE     STRING    = '1996-363-00:07:39.031'
N_SAMP (RADIOMETRIC) STRING    = '     512'
SAMP_INC (RADIOMETRIC) STRING    = '  23'
A3 (RADIOMETRIC) FLOAT     =       0.00000
Record sequence number:           2
Length of pixel count each line:       11968
Length of data pixel count each line:       11776
Acquisition date: 1996-363-      457.197sec
```

The following routine calls function sigma0.pro to convert the pixel values to calibrated sigma-naught. This routine requires knowledge of the starting pixel of the subsetted image in order to convert the sample number to the correct incidence angle.

```
PRO tosigma0,imin,js,imout
;; ================================================================
;; Given a image data and its starting range pixel js (corresponding
;; to the first column of the image array), convert the digital
;; number to Sigma Naught.
;;
;; Input:
;;    imin --- RSI image data, say a m x n image box
;;    js   --- starting range pixel for the 1st column of  imin
;;
;; Output:
;;    imout --- Sigma Naught for imin, save size as imin
;;
;; Oct. 22, 1999  Xiang Liu
;; ================================================================
sz=size(imin)
imout=fltarr(sz(1),sz(2))

for j=0,sz(2)-1 do begin
    for i=0,sz(1)-1 do begin      ;from near range to far range
        imout(i,j)=sigma0(js+i,imin(i,j))
    endfor
endfor

END


function sigma0,j,dn
;; ================================================================
;; This procedure calculates the Sigma Naught (radar backscatter
;; coeficient) in dB given the radar brightness value (Beta
;; Naught).
;;
;; Input:
;;      j  --- j-th pixel from the start of a range line in the
;;           detected image data
;;      dn --- the digital number which represents the magnitude
;;             of the j-th pixel
;;
;; Oct. 19, 1999  Xiang Liu
;; ================================================================
; get the Beta Naught
beta=beta0(j,dn)

; get the incidence angle at j th pixel
inc=incidence(j)

; get Sigma Naught
sigma0=beta+10*alog10(sin(inc))

return,sigma0
END
```

```
function beta0,j,dn
;; ================================================================
;; This procedure calculates the Beta Naught (radar brightness
;; value) in dB at j-th pixel.
;;
;; Input:
;;    j  --- j-th pixel from the start of a range line in the
;;           detected image data
;;    dn --- the digital number which represents the magnitude
;;           of the j-th pixel
;;
;; Created on Oct. 19, 1999  Xiang Liu
;; ================================================================

common data_sum,ellip_maj,ellip_min,pro_lat,plat_lat,pix_spacing
common proc_param,eph_orb_dat,srgr_coef
common radiometric,n_samp,samp_inc,lookup_tab,A3
common img_data,acc_year,acc_day,acc_msec,lat_mid

; extract fields from Radiometric Data Record
rd_rsi_tlr

; calculate a2
if (j le samp_inc*(n_samp-1)) then begin
   il=fix(float(j)/samp_inc)
   iu=nint(float(j)/samp_inc)
   ail=lookup_tab(il)
   aiu=lookup_tab(iu)
   a2=ail+((aiu-ail)*((float(j)/samp_inc)-il))
endif
if (j gt samp_inc*(n_samp-1)) then begin
   ail=lookup_tab(511)
   aiu=lookup_tab(510)
   a2=ail+((ail-aiu)*((float(j)/samp_inc)-511))
endif

; get Beta Naught
beta0=10.*alog10((float(dn)^2+A3)/a2)

return,beta0
END
```

```
function incidence,j
;; ====================================================================
;; This procedure calculates the incidence angle for the  RSI's
;; RadarSat ScanSar data.
;;
;; Oct. 19, 1999  Xiang Liu
;; ====================================================================

common data_sum,ellip_maj,ellip_min,pro_lat,plat_lat,pix_spacing
common proc_param,eph_orb_dat,srgr_coef
common radiometric,n_samp,samp_inc,lookup_tab,A3
common img_data,acc_year,acc_day,acc_msec,lat_mid

; read in the CEOS dataset summary record
rd_rsi_tlr

; calcualte the new platform latitude, only for ScanSAR products
newplat_lat=plat_lat+(pro_lat-lat_mid)

; earth radius (meter)
r=ellip_min*sqrt(1.+(tan(newplat_lat*!dtor))^2)*1000./   $
  sqrt(ellip_min^2/ellip_maj^2+(tan(newplat_lat*!dtor))^2)
;print,'Earth Radius:',r

; orbit altitude (meter)
h=eph_orb_dat-r
;print,'Orbit Altitude:',h

; calculate the slant range for each ground range increment
rs=srgr_coef(0)+j*pix_spacing*srgr_coef(1)+                        $
   (j*pix_spacing)^2*srgr_coef(2)+(j*pix_spacing)^3*srgr_coef(3)+  $
   (j*pix_spacing)^4*srgr_coef(4)*(j*pix_spacing)^5*srgr_coef(5)

; incidence angle
incidence=acos((h^2-rs^2+2.*r*h)/(2.*rs*r))

;print,'The Incidence Angle at '+string(j,format='(I5)')+ $
;      '-th pixel is',incidence*!radeg,' degrees'
return,incidence

END
```